

RTF Output Tips

- SAS初学者もSAS熟練者もたのしめるものを目指して -

○麻生 美由紀¹

(¹株式会社インテリム)

Practical Tips for RTF Output

- For Having RTF Output for ALL SAS Users to be Enjoyable -

Miyuki Aso

intellim Corporation

要旨

本論文では、SAS 初学者や RTF 出力を改めて学び直そうと考えている方に向けて、SAS での RTF 出力に関する学びのきっかけを掴んでもらうことを目的とし、基礎編・応用編に分けて RTF 出力に関する Tips を紹介する。まず、基礎編 (Basic Tips) として、RTF 出力に必要な要素として厳選した内容を実際の SAS プログラムを用いて解説する。具体的には、TEMPLATE プロシジャでのスタイルテンプレートの作成例、RTF 出力に関するオプションの指定例、ODS RTF ステートメントの使用例、REPORT プロシジャ及び ODS TABLE プロシジャの使用例を解説する。さらに、応用編 (Advanced Tips) として、日本語と英語が混在する RTF の帳票作成時に役立つ全角・半角フォントの制御方法について、筆者が SAS ユーザー総会 2018 で発表した内容の最新情報を紹介する。また、ODS RTF ステートメントとは異なる性質を持つ ODS TAGSETS.RTF ステートメントの使用例や RTF タグの活用事例を紹介する。本論文が SAS プログラマ間のさらなる意見交換や技術発展の一助になれば幸いである。

キーワード : RTF 出力, RTF Output, TEMPLATE プロシジャ, Style Template, スタイルテンプレート, ODS RTF ステートメント, ODS TAGSETS.RTF ステートメント, REPORT プロシジャ, ODS TABLE プロシジャ, フォント制御

1 はじめに

1.1 本論文の目的

本論文は SAS 初学者や RTF 出力の経験がない・少ない方或いは改めて学び直そうと考えている方に向けて、SAS での RTF 出力について学びのきっかけを掴んでもらうことを目的とし、基礎編・応用編に分けて RTF 出力に関する Tips を紹介する。

基礎編 (Basic Tips) では、実際に運用可能な SAS プログラムの例 (添付資料: ODS_RTF_sample_basic.sas) を通じて、RTF 出力に必要な要素を厳選して紹介する。

応用編 (Advanced Tips) では、SAS プログラムの例 (添付資料: ODS_RTF_sample_advanced.sas など) を通じて、RTF 出力での全角・半角フォントの制御方法、ODS TAGSETS.RTF ステートメントの使用例、RTF タグの活用事例など、RTF 出力に関する最新の技術や Tips を紹介する。

RTF 出力に関してより一層深い知識を得たい・プロシジャやステートメントの機能をもっと使ってみたいという方は、過去の SAS ユーザー総会の資料や SAS 社のマニュアルなどを活用いただきたい。

1.2 本論文で参照するプログラムと使用する SAS のバージョン

基礎編 (Basic Tips) では、ODS_RTF_sample_basic.sas を上から順に解説する。

応用編 (Advanced Tips) では、ODS_RTF_sample_advanced.sas 及び ODS_TAGSETS.RTF_sample.sas を参照する。RTF タグ活用事例は紹介に留める。


また、本論文で使った SAS のバージョンは SAS 9.4M7 (TS1M7) である。

2 Basic Tips

本章では、RTF 出力の基礎を ODS_RTF_sample_basic.sas の SAS プログラムコードを通じて紹介する。ここでは RTF 出力に必要な要素の説明に留め、各ステートメントや各プロシジャの詳細な説明は割愛する。基礎編のプログラムを参考に、自らプログラム作成する、プログラムの挙動を確認する、分からない部分は SAS 社のマニュアル [1] や論文を検索・参照するなどして、各ステートメントや各プロシジャの機能を理解していただきたい。

ここで、SAS における RTF 出力までの基本的な流れを図 2-1 に示す。

図 2-1 SAS における RTF 出力までの基本的な流れ



RTF 出力の基本的な流れを一通り経験する目的で、RTF 出力までの全工程を 1 つのプログラムに含めたものを紹介する。なお、ODS_RTF_sample_basic.sas は学習用のサンプルプログラムであり、実業務での運用は想定していない。このプログラムはあくまで参考として、各自の所属の会社・組織の方針に従い、各自で適切なプログラムを作成していただきたい。

2.1 TEMPLATE プロシジャを用いたスタイルテンプレートの作成例

TEMPLATE プロシジャで作成したスタイルテンプレートを ODS RTF ステートメントの STYLE オプショ

ンにおいて指定することで、フォントの設定や文字色の設定、罫線の設定、余白の設定などの RTF 出力の見た目を制御することができる [2]。ODS RTF ステートメントの STYLE オプションを指定しなくても RTF 出力は可能だが、大抵の場合、見た目の微調整が必要である。スタイルテンプレートを作成しておくことで、出力の全体的な見た目を制御できるため、各レポート作成用のプログラムにおいてスタイルを細かく指定する必要がなくなり、大変便利である。


まず、ODS PATH ステートメントでのスタイルテンプレートの保存先・参照先の変更について解説する。

CODE 1: ODS PATH Statement

```
ods path reset;  
ods path(remove) sasuser.templat;  
ods path(prepend) work.templat(update);
```

通常、新しく作成するテンプレートは SASUSER.TEMPLAT に保存される。CODE 1 では、ODS PATH ステートメントによってスタイルテンプレートの保存先を WORK.TEMPLAT (WORK ライブラリ内のスタイルテンプレートの保存先) に変更している。現在の ODS PATH は、[ods path show;]のコードを実行すると確認可能で、SAS のログ画面に図 2-2 のように表示される。

図 2-2 CODE 1 を実行した際の ODS PATH (左) とデフォルトの ODS PATH (右) を示すログ画面



次に、ODS_RTF_sample_basic.sas における TEMPLATE プロシジャの使用例を解説する。

CODE 2: TEMPLATE Procedure

```
proc template;  
  define style Myrtftemp;  
    parent=styles.RTF;  
    style systemtitle/  
      font_face = "Times New Roman"  
      font_size = 9pt  
      font_weight = medium  
      font_style = roman  
      foreground = black  
      background= white;  
  (中略)  
  end;  
run;
```

CODE2 では、TEMPLATE プロシジャの DEFINE STYLE ステートメントで、新たにスタイルテンプレート

Myrtftemp を作成している。作成した Myrtftemp は、上述の CODE 1 の ODS PATH の設定によって、WORK.TEMPLAT 内に保存される。

スタイルテンプレートを新規に定義するのは難しいため、ここでは PARENT=ステートメントで、SASHELP.TMPLMST 内に保存されている styles.RTF を親スタイルテンプレートに指定して Myrtftemp を作成している。DEFINE STYLE ステートメントから end; までの DEFINE STYLE ステートメントブロック内において、STYLE ステートメントまたは CLASS ステートメントでスタイル要素を加工することで、スタイルテンプレートをカスタマイズしている。従前より REPLACE ステートメントがよく使われているが、現在、REPLACE ステートメントはサポートされておらず、TEMPLATE プロシジャのマニュアルでは STYLE/CLASS ステートメントの利用が推奨されている [3]。

ここで、スタイル要素 (style elements) とスタイル属性 (style attributes) について説明する。

スタイル要素は、ヘッダー/フッター、表ヘッダー、表フッター、表中の値、などの文書スタイルを構成する要素を指し、それぞれのスタイル要素に属するものとして、スタイル属性が存在する。CODE 2 に示す例では、STYLE ステートメントで指定している systemtitle がスタイル要素であり、font_face=などがスタイル属性に該当する。どのようなスタイル要素・スタイル属性が指定可能かを確認する方法は、SAS ヘルプセンター内のマニュアル[4][5][6]を参照する方法や過去の論文などを参照する方法が有効である。これらの方法は、特に SAS 初学者にとっては検索のハードルが高いと思われるため、ODS_RTF_sample_basic.sas の TEMPLATE プロシジャ内に、比較的頻繁に指定するスタイル要素とスタイル属性を例示している。なお、スタイルテンプレートの中身を知りたい場合、[proc template; source Myrtftemp; run;] を実行すると、ログ上にスタイルテンプレートの詳細な内容が表示される。また、SASHELP.TMPLMST や WORK.TEMPLAT 内に保存されているスタイルテンプレートを確認したい場合は、[proc template; list styles / store=sashelp.tmplmst; run;] を実行すると、アウトプット画面にスタイルテンプレート一覧が表示される。

2.2 OPTIONS/TITLE/FOOTNOTE ステートメントの使用例

CODE 3: OPTIONS Statement

```
options nodate nonumber;  
options orientation="portrait";
```

まず、RTF 出力における OPTIONS ステートメントの役割を紹介する。CODE 3 の options nodate nonumber; は、SAS Output に表示されるシステム日付及びページ番号を抑制するために指定している。また、options orientation=では、ページの縦 (Portrait)・横 (Landscape)を指定している。RTF 出力ではページ縦表示か横表示かで、出力可能な列数や出力可能な列の幅の合計が異なってくるので、レイアウト作成段階に、ページの縦・横をあらかじめ決めておくことが肝要である。

CODE 4: TITLE Statement and FOOTNOTE Statement

```
title1 j=right "Basic Tips";  
title2 j=right "ODS RTF Sample (proc report)";  
footnote1 j=center "Page ~{thispage} of ~{lastpage}";
```

次に、RTF 出力における TITLE ステートメント及び FOOTNOTE ステートメントについて紹介する。TITLE/FOOTNOTE ステートメントに指定した文字列は、RTF 出力のヘッダー/フッター部分に表示される

(ODS RTF ステートメントで BODYTITLE オプションを指定しない場合)。また、どちらのステートメントにも文字列の表示位置を指定する justify=(j=)オプションが存在する。CODE 4 では、タイトルは右寄せ (right)、フットノートは中央揃え (center) で配置されるように指定している。TITLE/FOOTNOTE ステートメントには、他にも文字色、太字、フォント等を指定するオプションが存在するが、ODS_RTf_sample_basic.sas では、既にスタイルテンプレートにおいて、ヘッダー/フッターのスタイルを定義しているため (2.1, 2.4 節を参照)、TITLE/FOOTNOTE ステートメントでのオプション指定は不要である。なお、TITLE と FOOTNOTE は最大 10 個まで指定できる [7]。

また、CODE 4 の footnote1 に示す通り、エスケープ文字 (2.3 節を参照) を利用して、[エスケープ文字(~)+{thispage}]で現在のページ番号、[エスケープ文字(~)+{lastpage}]で総ページ数を表示することができる。~{thispage}や~{lastpage}は、ODS RTF ステートメントのオプションで BODYTITLE オプションを利用しない場合のみ有効であり、ODS RTF ステートメントのオプション BODYTITLE を指定した場合には、ページ数が表示されないので注意が必要である。

2.3 ODS ESCAPECHAR ステートメントの指定例

ODS ESCAPECHAR ステートメントは、インラインフォーマット関数を利用するためのエスケープ文字を指定するステートメントである。CODE 5 の ods escapechar=では、エスケープ文字として「~」(チルダ) を指定している。RTF 出力の場合は、エスケープ文字+インラインフォーマット関数を文字列の先頭に指定することで、文字列のスタイルを変更することができる。また、セル内改行、上/下付き文字、ページ番号などを RTF 出力内に入れる際にも、エスケープ文字は用いられる。3.1 節で紹介する RTF 出力での全角・半角フォントの制御方法においても、エスケープ文字の指定が重要になる。エスケープ文字に使用可能な文字やインラインフォーマット関数の詳細については、SAS ヘルプセンター内のマニュアルを参照されたい [8]。

CODE 5: ODS ESCAPECHAR Statement

```
ods escapechar="~";
```

2.4 ODS RTF ステートメントの使用例

ODS RTF ステートメントは、Microsoft Word で使用する RTF 形式の出力を制御するためのステートメントである。また、ODS RTF ステートメントには CODE 6 で紹介する FILE=/ STYLE=/ STARTPAGE=/ BODYTITLE/ TEXT=/ PREPAGE=オプションの他にも多数のオプションが用意されているため、関心のある方は確認されたい [9]。

CODE 6: ODS RTF Statement

```
ods rtf file="&sasprgpath.¥Basic_procreport.rtf" style=Myrtftemp startpage=yes ;
```

ODS RTF ステートメントの FILE=オプションによって、RTF 出力先 (書き込み先) の RTF ファイルが開かれ、[ods rtf close;] または[ods _all_ close;]の操作を行うまで開いたままになる。STYLE=オプションにて、2.1 節の CODE 2 にて作成したスタイルテンプレート Myrtftemp を指定し、RTF 出力に定義済みのスタイルを適用する。STARTPAGE=オプションは改ページを制御するオプションで、YES の場合はプロシジャごとに改ページされる。STARTPAGE=YES がデフォルトの設定となっている。プロシジャごとに改ページしたくない場合は STARTPAGE=NO を指定し、改ページしたい場所に[ods rtf startpage=now;]を記述する。

なお、CODE 6 には記載していないが、ODS RTF ステートメントには BODYTITLE というオプションが存在する。BODYTITLE を指定すると、TITLE/FOOTNOTE ステートメントで指定したテキストが RTF 出力の本文中に出力される。一方、CODE 6 では BODYTITLE を指定していないため、TITLE/FOOTNOTE ステートメントで指定したテキストはそれぞれ RTF 出力のヘッダー/フッターに出力される。

CODE 7: ODS RTF PREPAGE= / TEXT= Option

```
ods rtf prepage="Listing16.2.1 SASHELP.CLASSFIT";  
[PROC REPORT の出力1]  
ods rtf text="This is a text like footnote.";  
ods rtf prepage="Listing16.2.2 SASHELP.CLASSFIT";  
[PROC REPORT の出力2]  
ods rtf text="This is a text like footnote.";
```

次に、任意のテキストを RTF 出力に表示するためのオプションの使用例を示す。ODS RTF PREPAGE=は RTF 出力の各プロシジャで作成される表の前に挿入される文字列を指定するために用いられる。一方、ODS RTF TEXT= (ODS TEXT=)は RTF 出力の任意の場所に文字列を挿入する際に用いられる。文字列を挿入するという点では似ているが、厳密には微妙に機能が異なる。ODS RTF ステートメントで STARTPAGE=YES を指定した場合、ODS RTF PREPAGE を [PROC REPORT の出力 1]、 [PROC REPORT の出力 2] の間に記述したとき、ODS RTF PREPAGE で指定したテキストは [PROC REPORT の出力 2] のタイトルのように出力される。一方、ODS RTF TEXT を [PROC REPORT の出力 1]、 [PROC REPORT の出力 2] の間に記述したとき、ODS RTF TEXT で指定したテキストは [PROC REPORT の出力 1] の脚注のように出力される。これらの文字列を挿入するオプションは、TITLE/FOOTNOTE ステートメントの代替として、表タイトルや表脚注を挿入する際に用いられる。ただし、ODS RTF STARTPAGE=NO か YES の違いで、表とテキストの間の空白行の有無が異なるため、注意が必要である。

2.5 REPORT プロシジャと ODS RTF プロシジャの使用例

本節では、RTF 出力の表作成に使用される代表的なプロシジャである、REPORT プロシジャ・ODS RTF プロシジャの使用例を紹介する。ODS RTF_sample_basic.sas においては、2.1 節の CODE 2 にて作成したスタイルテンプレートを利用するため、プロシジャ内のオプションやインラインフォーマット関数などを用いたスタイルの設定は最小限に留めている。CODE 8 及び CODE 9 は、SASHELP.CLASSFIT の一部の変数を計算・加工して作成した OUT1 から OUT5 までの文字値変数を持つデータセットを出力するための REPORT プロシジャ・ODS RTF プロシジャのコードである。どちらのコードも同様の見た目の出力を作成できる。

CODE 8: REPORT Procedure

```
%let COMMON_BTMBORDER=%str(borderbottomwidth=1pt borderbottomcolor=black borderbottomstyle=solid);  
  
proc report data=_classfit missing split=#;  
  column OUT1-OUT3 ("S={&COMMON_BTMBORDER.}Physique" OUT4 OUT5);  
  define OUT1/ display "Name" style(header)=[just=left] style(column)=[cellwidth=40mm just=left];  
(中略 ★下に続く)
```


(中略 ★上の続き)

```
compute after _page_/style=[just=left vjust=top borderbottomstyle=none];  
line "Customize footer by yourself."  
endcomp;  
run;
```

REPORT プロシジャには、表作成機能 (PRINT プロシジャとほぼ同様の機能) の他にも様々な機能が搭載されており、オプションもそれぞれのステートメントに対して多く用意されているが、本論文で紹介する機能は、表作成機能+ α に絞っている。なぜなら、特に統計解析業務においては、REPORT プロシジャを用いて要約・計算・データ加工などを行うことはほとんどなく、出力用 (解析結果表示用) のデータセットを予め手前の解析プロシジャ及びデータステップで作成しておくことが一般的であり、実際に使用する機能は限られているからである。REPORT プロシジャのその他の機能については、SAS ヘルプセンター内のマニュアルや過去の論文等を参照されたい [10] [11]。

CODE 8 では、COLUMN ステートメント内で、複数の変数を括弧で囲み、OUT4 と OUT5 に共通の表ヘッダーを追加している。この表ヘッダーのテキストには、インラインフォーマット関数(CODE 8 の%let COMMON_BTMBORDER= 及び ~S={&COMMON_BTMBORDER.})の部分(参照)で下罫線を引く指定をしている。このため、図 2-3 のような表示になる。

図 2-3 CODE 8 の COLUMN ステートメントで表ヘッダーを追加したときの RTF 出力画面




Name	Sex	Age (years)	Height (cm)	Weight (kg)
Joyce	F	11	130.3	22.9
Louise	F	12	143.0	34.9

また、COLUMN ステートメントで宣言した変数について、DEFINE ステートメントで出力用に作成されたデータセットの変数をそのまま表示 (DISPLAY オプションを指定) している。DEFINE ステートメントでは、データセット内の変数ごとに、変数の役割や表ヘッダーとなるテキスト、スタイルを指定することができる。CODE 8 の DEFINE COL1/ の部分では、変数の役割を DISPLAY とし、STYLE(HEADER)=オプションでは表ヘッダーのスタイル属性 (just=left) を指定、STYLE(COLUMN)=オプションでは、表中の値とセル幅などのスタイル属性 (cellwidth=40mm just=left) を指定している。変数ごとに、文字の配置、文字色や背景色などを変更したい場合は、これらの変数ごとの個別のオプションが有効である。

また、compute after _page_ から endcomp; までの COMPUTE 構文の中で LINE ステートメントを用いることで、図 2-4 に示す通り表の脚注として文字列を追加することができる。さらに、COMPUTE ステートメントを用いれば、データセットの変数の値ごとにスタイルを変更 (例えば、特定の位置で下罫線を引くなど) が可能である。

図 2-4 CODE 8 の COMPUTE ステートメントで表の脚注を追加したときの RTF 出力画面



Alfred	M	14	175.3	51.0
Philip	M	16	182.9	68.0
Customize footer by yourself.				

CODE 9: ODSTABLE Procedure

```
proc odsttable data=_classfit;
  column OUT1-OUT5;
  define header hd45; text "Physique"; start=OUT4; end=OUT5; style={&COMMON_BTMBORDER.}; end;
  define OUT1;
    define header hd1; text "Name"; style={just=left}; end; style={cellwidth=40mm just=left}; header=hd1;
  end;
  (中略)
  define footer btm1; text "Customize footer by yourself."; start=OUT1; end=OUT5; style={just=left vjust=top}; end;
run;
```

ODSTABLE プロシジャは、TEMPLATE プロシジャの DEFINE TABLE ステートメントと同様の機能を持ち、表形式の出力テンプレートを作成することに特化したプロシジャである。表を含む RTF 出力を作成する場合に多く利用されている。

CODE 9 では、COLUMN ステートメントで宣言した変数について、DEFINE ステートメントにおいてそれぞれ定義をしている。また、DEFINE HEADER、DEFINE FOOTER では、それぞれ表ヘッダーと表フッターの定義を行っている。表ヘッダーと表フッターも DEFINE ステートメントで定義できることが ODSTABLE プロシジャの特徴である。CODE 9 では、define header hd45; の部分で hd45 という表ヘッダーを定義し、start=OUT4; end=OUT5; とすることで、OUT4 と OUT5 に共通の表ヘッダーを追加している。これによって、REPORT プロシジャでの、COLUMN ステートメント内で複数の変数を括弧で囲む方法と同じ結果となる (図 2-3 と同様の結果となる)。define footer btm1;の部分では、btm1 という表フッターを定義し、start=OUT1; end=OUT5; とすることで、表フッターを追加している。これによって、REPORT プロシジャでの、compute after _page_ から endcomp;までの COMPUTE 構文の中で LINE ステートメントを用いて表の脚注のような文字列を追加する方法と同じ結果となる (図 2-4 と同様の結果となる)。なお、ODSTABLE プロシジャでのスタイルの定義の仕方は、REPORT プロシジャと類似しているため、詳細はプログラム内のコードを参照されたい。

RTF 出力での表作成の場合、各社・各組織の実業務で運用する上では、REPORT プロシジャと ODSTABLE プロシジャのどちらを使うかをプログラマ間で統一することを勧めたい。なぜなら、どちらのプロシジャを用いても工夫次第で同一の表現が可能とはいえ、そもそもの仕組みが異なるプロシジャであり、プログラマが各々別のプロシジャで表を作成すると、インラインフォーマット関数の指定箇所や表ヘッダー/表フッターの構成などの違いによって RTF タグの構成が異なり、後工程で RTF ファイルを比較することが困難になるからである。どちらのプロシジャが、或いは他のプロシジャが、より自分たちが表現したい内容を適切に表現できるかを十分に検討したうえで、プロシジャの使い分けとプログラマ間での使用プロシジャの統一を推奨する。

3 Advanced Tips

本章では、応用編として RTF 出力において有用な技術・Tips を紹介する。3.1 節で紹介するフォント制御の他にも応用的な技術を取り入れたプログラム ODS_RTF_sample_advanced.sas を用意した。基礎編同様、詳細な説明は省くが、実際に挙動を確かめることで、ある程度理解は可能である。本章で紹介する内容が、RTF

出力の可能性を広げる一助となれば幸いである。

3.1 RTF 出力での全角・半角フォントの制御方法

日本語と英語が混在する RTF の帳票作成時に役立つ全角・半角フォントの制御方法について、筆者が SAS ユーザー総会 2018 にて発表した方法の最新版を紹介する。

まず、全角・半角フォントの制御方法を開発するに至った背景を説明する。日本語と英語が混在している帳票において、「日本語：MS 明朝、英数字：Times New Roman」「日本語：MS ゴシック、英数字：Arial」などの組み合わせを依頼された場合、現状 (2024 年現在も)、TEMPLATE プロシジャで作成するスタイルテンプレートや REPORT プロシジャの STYLE オプションなどに、全角文字及び半角文字表示用のフォントを個別に指定できるオプションは存在しない。そこで筆者は、RTF ファイルを Word で開き直してフォントをかけ直すといった処理などをせずに、1 回の出力で、なおかつ RTF タグの加工もせずに、RTF 出力の全角・半角文字について任意のフォントを制御できる方法を開発し、SAS ユーザー総会 2018 にて発表した [13]。本論文では、SAS ユーザー総会 2018 で紹介したコードよりも簡単なコードを提案する。

全角・半角フォントの制御方法を利用する上での前提条件は、ODS RTF ステートメントを使用して RTF 出力を作成することと、2.3 節で紹介したエスケープ文字 (例示では「~」チルダ) を定義しておくこと、以上 2 つである。スタイルテンプレートにて定義するフォントは何でも良いが、全角文字が現れるときにだけ全角・半角フォントの制御をすれば良いので、半角文字表示用のフォント (例えば、Times New Roman) を指定しておくことを推奨する。

まず、全角・半角フォントの制御方法の基本の形を紹介する。インラインフォーマット関数でのフォントの指定を重ねて 2 回行う (~S={font_face='MS 明朝'}~S={font_face='Times New Roman'}) ことである。全角文字表示用フォント→半角文字表示用フォントの順で指定することがポイントである。

CODE A1: 全角・半角フォントの制御方法の基本の形

```
ods rtf prepage="~S={font_face='MS 明朝'}~S={font_face='Times New Roman'}Listing16.2 SASHELP.CLASS (全角半角フォント制御あり)";
```

次に、REPORT プロシジャでの全角・半角フォント制御方法を紹介する。

CODE A2: REPORT プロシジャでの全角・半角フォント制御の例

```
proc report data=_class missing split=#'
  style(header)={pretext="~S={font_face='MS 明朝'}~S={font_face='Times New Roman'}"}
  style(column)={pretext="~S={font_face='MS 明朝'}~S={font_face='Times New Roman'}"}
  style(lines)={pretext="~S={font_face='MS 明朝'}~S={font_face='Times New Roman'}"};
  column OUT1-OUT3 ("~S={&COMMON_BTMBORDER;}体格" OUT4 OUT5);
  define OUT1/ display "名前" style(header)={just=left} style(column)={cellwidth=40mm just=left};
(中略)
run;
```

CODE A2 では、REPORT プロシジャの PROC REPORT ステートメントの STYLE オプションにて、各 location (header, column, lines) に対して {pretext="~S={font_face='MS 明朝'}~S={font_face='Times New Roman'}"} を指定している。このように pretext= を用いる方法によって、SAS ユーザー総会 2018 にて紹介した方法 (REPORT

プロシジャの COMPUTE ステートメントを用いて内部的にデータにインラインフォーマット関数を付与し、スタイルをデータセット内の文字列に対して直接制御する方法) よりも比較的簡単に全角・半角フォント制御が可能になった。ODSTABLE プロシジャでも同様に `pretext=`を用いる方法が有効である。ODSTABLE プロシジャでの全角・半角フォント制御例は `ODS_RTF_sample_advanced.sas` 内のコードを参照されたい。

3.2 ODS TAGSETS.RTF ステートメントの使用例

ODS TAGSETS.RTF ステートメントとは、2.4 節で紹介した ODS RTF ステートメントと同様に、RTF 形式の出力を制御するためのステートメントである [14][15]。ただし、ODS RTF ステートメントとは異なり、「Microsoft Word で使用する」という前提はなく、ODS TAGSETS.RTF ステートメントで作成する RTF 出力は、Microsoft Word に依存しない RTF 出力 (Measured RTF) となるのが特徴である。また、ODS RTF ステートメントとは種類の違うステートメントであり [16]、ODS RTF を ODS TAGSETS.RTF に置き換えただけでは上手く機能しない場合がある。加えて、ODS RTF と比べて RTF タグの構成も異なっている。

また、ODS TAGSETS.RTF ステートメントで生成される Measured RTF の特徴を以下に挙げる。

- Microsoft Word 以外の RTF リーダーで読み取ることができる
- 非常に大きな表の改ページを (メモリ消費量を節約しながら) 制御できる
- タイトル・フットノートを本文中に表として配置する

Measured RTF では、Microsoft Word 特有のヘッダー/フッターを使用しないため、ページ番号(`~{thispage}`や`~{lastpage}`) が使えないという難点があるが、RTF 出力自身が Microsoft Word に依存せずに、改ページやタイトル・フットノートの配置を決められるという利点がある (例えば、非常に大きな表を出力する場合、ODS RTF で作成される従来の RTF は表全体を読み込んでから Microsoft Word によって改ページの処理をしており、ページ間で情報を維持するために各ページの出力の際は多くのメモリを必要とするのに対して、Measured RTF は SAS のタグセットによってあらかじめ改ページ位置を決めているので各ページの出力の際は 1 ページ分のメモリのみで済む)。

筆者は、ODS TAGSETS.RTF ステートメントが、今後 ODS RTF ステートメントに取って代わるのか、どのような点が実業務上で役に立つかなどを検証するために、ODS TAGSETS.RTF ステートメントを用いて RTF 出力を作成するプログラム `ODS_TAGSETS.RTF_sample.sas` を作成した。以下に、ODS TAGSETS.RTF ステートメントの使用例と ODS RTF ステートメントとの違いなどを紹介する。

CODE A3: ODS TAGSETS.RTF Statement

```
ods tagsets.rtf file="&sasprgpath.¥Advanced_odstagsetsrtf_procreport.rtf"
               style=Myrtftemp
               startpage=yes
               options(vspace='off');
ods tagsets.rtf options(tables_off='prepage');
ods tagsets.rtf prepage="Listing16.2.1 SASHELP.CLASSFIT";
```

まず、ODS TAGSETS.RTF ステートメント固有の機能を紹介する。CODE A3 の `OPTIONS(VSPACE='OFF')` オプションは、タイトルと表の間の空白及び表とフットノートの間の空白を取り除くためのオプションである。一方、`VSPACE='ON'` とすれば、ODS RTF ステートメントの出力と同様に空白を加えることができる(図 3-1 参照)。 `OPTIONS(TABLES_OFF='PREPAGE')` オプションは、引用符の中にスタイル要素名を指定するこ

とで、そのスタイル要素の表出力をオフにする (本文中に直接書き込む) ことが可能になる。

その他にも ODS TAGSETS.RTF ステートメント固有のオプションがあるので、詳細は SAS ヘルプセンター内のマニュアルを参照されたい。ODS TAGSETS.RTF には ODS RTF と共通しているオプションも存在するが、ODS RTF では使用可能だが ODS TAGSETS.RTF では使用不可の場合 (あるいはその逆) やコードの記載の仕方が違う場合があるので、使用の際はよく確認してほしい。

図 3-1 ODS TAGSETS.RTF OPTIONS(VSPACE='OFF') (上) と OPTIONS(VSPACE='ON') (下)の違い

Advanced Tips ODS TAGSETS.RTF Sample (proc report)				
Listing16.2.1 SASHELP.CLASSFIT				
			Physique	
Name	Sex	Age (years)	Height (cm)	Weight (kg)
Joyce	F	11	130.3	22.9

次に、ODS RTF ステートメントとの違いを紹介する。特に大きな違いが発生するのは、TITLE/FOOTNOTE ステートメントの挙動である。TITLE/FOOTNOTE ステートメントで指定したテキストは、ODS RTF (BODYTITLE 指定なし) の場合はヘッダー/フッター中のテキストとして出力され、各ページにまたがって表示されるが、ODS TAGSETS.RTF の場合は本文中のタイトル・フットノートとして出力され、各ページにまたがって表示される。また、ODS RTF (BODYTITLE または BODYTITLE_AUX 指定あり) の場合、本文中に出力されるが、表が 1 ページに収まらない場合に、表の一番上と一番下に表示されるのみである。一方、ODS TAGSETS.RTF の場合は、表が 1 ページに収まらない場合に、各ページに分割された表の上と下に、常にタイトル・フットノートとして表示される。

また、ODS TAGSETS.RTF での改ページの仕方、PREPAGE/TEXT ステートメントの出力位置についても、ODS RTF とは違う挙動を示す。表作成に使用するプロシジャが REPORT プロシジャか ODSSTABLE プロシジャかでも違いが生じる場合がある。出力位置や出力方法の違いについて詳細を確認したい場合は、それぞれのステートメントで作成した出力結果を見比べて確認することを推奨する。

以下は筆者の所感であるが、ほとんどの RTF 出力作成時において、ODS RTF ステートメントの代わりに ODS TAGSETS.RTF ステートメントをわざわざ使う必要はないと考える。ただし、非常に大きな表で各ページに出力する行数を決めたい場合 (TABLEROWS=オプションを使う) や表がページをまたぐ場合に継続タグをつけたい場合 (OPTIONS(CONTINUE_TAG='ON')オプションを使う) など、ODS TAGSETS.RTF ステートメントならではの機能を十分に理解できれば、効果的に使用可能と考える。実業務での実際の運用に至るかどうか、検討する余地は残されている。今後の研究課題として本論文で紹介した内容が参考になれば幸いである。

3.3 RTF タグの活用事例

RTF ファイルをメモ帳で開くと図 3-2 のように表示でき、RTF タグを確認できる。

図 3-2 SAS で出力した RTF ファイルをメモ帳で開いた時の様子

```

ファイル(F) 編集(E) 書式(O) 表示(V) ヘルプ(H)
{#rtf1#ansi#ansicpg932#uc1#deff0#deflang1041#deflangfe1041
{#fonttbl
{#f1#fmodern#fprq1#fcharset128#cpg932 Courier;}
{#f2#froman#fprq2#fcharset128#cpg932 MS PMincho;}
{#f3#froman#fprq2#fcharset128#cpg932 Times New Roman;}
}{#colortbl;
#red0#green0#blue0;
#red0#green0#blue255;
#red0#green255#blue255;
#red0#green255#blue0;
#red255#green0#blue255;
#red255#green0#blue0;
#red255#green255#blue0;
}

```

SAS で出力した RTF ファイルの RTF タグは比較的シンプルな構成をしているため、RTF ファイルを SAS データセットとして再読み込みすることで、RTF タグの加工や RTF ファイル同士の結合、RTF タグを利用した RTF ファイルのコンペアなどが可能になる。ただし、SAS で出力した RTF ファイルを Microsoft Word で開き再保存した場合は、RTF タグに余分な情報が付加され、非常に複雑な RTF タグ構成を持つ RTF ファイルとなるので注意が必要である。さらに、RTF タグの構成を理解しないまま、RTF タグ加工や RTF ファイル結合などをすると、RTF ファイルを開くことができなくなる (RTF ファイルが破損する) 恐れがあるので注意が必要である。また、一度出力した結果に対して加工することになるため、実業務での運用は結果の信頼性の観点から憚られるかもしれない。ここでは、業務効率化の一環として RTF タグを活用した事例を 2 つ紹介する。ただし、実業務における運用事例ではなく、開発中の内容を含み、あくまで活用の可能性を検討しているものをご承知いただきたい。

【ODS RTF WORDSTYLE=オプションを利用したオリジナルの見出し 1 の作成】

ODS RTF WORDSTYLE=オプションを用いると Microsoft Word 上のみで使用可能な見出しの情報を RTF ファイルに付与することができる。使用例を CODE A4 に示す。CODE A4 では引用符の中に{¥s1 heading 1;}とあるが、これによって、¥s1 (RTF のマークアップ用のタグ) を「見出し 1」と認識させることができる。

¥s1 はエスケープ文字と併せて使用し、見出し 1 と認識させたいテキストの先頭に付加する。その後、一旦出力した¥s1 のタグを含んだ RTF ファイルを SAS データセットで再び読み込み、¥s1 によって「見出し 1」と認識されたテキストに対して本文中に直接書き込むように RTF タグの加工を行うと、図 3-3 のように見出し 1 のテキストがナビゲーションウィンドウに表示されるようになる。この方法は、オリジナルの目次 (Table of Contents) を作成する際にも利用することができる。


CODE A4: ODS RTF WORDSTYLE= Option and ODS RTF PREPAGE= Option

```

ods rtf file="xxxxxxx.rtf"
    wordstyle={¥s1 heading 1.};
ods rtf prepage=" ~¥s1 Listing16.2.1 SASHELP.CLASSFIT";
(中略)
ods rtf prepage=" ~¥s1 Listing16.2.2 SASHELP.CLASSFIT";

```

図 3-3 ODS RTF PREPAGE=で指定したテキストを見出し 1 として表示したときの Word 上の画面



【RTF ファイルの結合】

個別に出力した RTF ファイル同士の結合を行うための SAS プログラムを開発した。プログラムの流れは以下の通りである。

- 結合元の RTF ファイル (No.1) を SAS データセットとして読み込んで末尾の RTF タグを加工する。
- 結合する RTF ファイル (No.2) を SAS データセットとして読み込んで結合に必要な部分だけを抽出し、RTF タグを加工する。
- No.1 の加工済 SAS データセットと No.2 の加工済 SAS データセットを縦結合する。
- 上述の工程を結合する RTF ファイルの数だけ繰り返し、最終的に出来上がった SAS データセットを RTF ファイルとして再出力する。

ただし、同じスタイル定義・同じ環境で作成した RTF ファイルの結合に限られる、という制限がある。RTF ファイルの結合についてはその他にも方法が報告されているが、この方法は、Word や VBA などの機能を使用せずに、SAS の機能のみで実現しているという点で画期的である。

4 最後に

本論文の Basic Tips を参考にすれば基本的な RTF 出力が一通り実現可能になると考えている。ただ、ここで紹介した技術や Tips が決して RTF 出力の最適解ではなく、まだまだ改良の余地があると思う。プロシージャに搭載された機能をより効果的に使えば、より多様な表現ができるかもしれない。サブタイトルに記載しているように、紹介した技術や Tips が、全 SAS ユーザーにとってプログラム開発の手助けとなるとともに、プログラム開発を”たのしむ”為のツールとなることを期待している。各プログラマが試行錯誤しながらツールを引用・改良することで、自分が納得できるプログラムを開発することを、是非たのしんでいただきたい。また、そのようにプログラム開発する中で発見されるより良い技術や Tips の共有が、今後も活発になされていくことを期待する。

5 参考文献

- [1] SAS® Output Delivery System: Procedures Guide. SAS® Help Center (Programming Documentation for SAS® 9.4 and SAS® Viya® 3.5).
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsproc/titlepage.htm
- [2] TEMPLATE Procedure: Creating a Style Template - Example 3: Modifying the Default Style with the CLASS Statement. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsproc/p1trs6h1rlpxgjn17yofb92v31xz.htm
- [3] TEMPLATE Procedure: Creating a Style Template - REPLACE Statement. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsproc/p126rcnu6y4s8cn1sg6i4sdq9g6i.htm
- [4] ODS Style Elements. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsproc/p0pm4ysu0fb68dn1d8uj78weie6o.htm
- [5] Style Attributes Tables. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsproc/n0otdo2g12obp3n0zmnghcn7p4vu.htm
- [6] Style Attributes Detailed Information. SAS® Help Center.

- https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsproc/p1pt77toue3iyun0z4I9gth5as9f.htm
- [7] TITLE Statement. SAS® Help Center.
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/lestmtsglobal/p10gcmmf83iaxn1ilrx4pra969n.htm
- [8] ODS ESCAPECHAR ステートメント. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsug/p11xia2ltavr8ln17srq8vn4rnqc.htm
- [9] ODS RTF ステートメント. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsug/p1vvsv8ucnjzjnn1wq5wrlp74mdb.htm
- [10] REPORT プロシジャ. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/proc/p0bqogcics9o4xn17yvt2qjbgdpi.htm
- [11] 太田裕二. 2017. ODS 機能と REPORT プロシジャを用いた 解析帳票の作成. SAS ユーザー総会 2017.
https://www.sas.com/content/dam/SAS/ja_jp/doc/event/sas-user-groups/usergroups2017-b-11-02.pdf
- [12] The ODSTABLE Procedure. SAS® Help Center.
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/odsproc/n1olg2oeqi9r5m1racah27f1ow8.htm
- [13] 麻生美由紀. 2018. ods rtf+proc report における解析帳票の日本語/英数字フォントの明示的制御. SAS ユーザー総会 2018.
https://www.sas.com/ja_jp/events/18/users-group-2018/sessions.html#m=4-2
- [14] ODS TAGSETS.RTF Statement. SAS® Help Center.
https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/odsug/p19rpsb989jyadn1kgunyh4t8g0e.htm
- [15] Xiangxiang Meng. 2010. ODS TAGSETS.RTF: Tips and Tricks. MWSUG-2010-67.
<https://www.mwsug.org/proceedings/2010/dataviz/MWSUG-2010-67.pdf>
- [16] ODS の仕組みの概要. SAS® Help Center.
https://documentation.sas.com/doc/ja/pgmsascdc/9.4_3.5/odsug/p027b3wsp0yg9yn1tfqekatdtn85.htm